

## F08QLF (STRSNA/DTRSNA) – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

F08QLF (STRSNA/DTRSNA) estimates condition numbers for specified eigenvalues and/or right eigenvectors of a real upper quasi-triangular matrix.

### 2 Specification

```

SUBROUTINE F08QLF(JOB, HOWMNY, SELECT, N, T, LDT, VL, LDVL, VR,
1             LDVR, S, SEP, MM, M, WORK, LDWORK, IWORK, INFO)
ENTRY      strsna(JOB, HOWMNY, SELECT, N, T, LDT, VL, LDVL, VR,
1             LDVR, S, SEP, MM, M, WORK, LDWORK, IWORK, INFO)
INTEGER    N, LDT, LDVL, LDVR, MM, M, LDWORK, IWORK(*), INFO
real     T(LDT,*), VL(LDVL,*), VR(LDVR,*), S(*), SEP(*),
1             WORK(LDWORK,*)
LOGICAL    SELECT(*)
CHARACTER*1 JOB, HOWMNY

```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

This routine estimates condition numbers for specified eigenvalues and/or right eigenvectors of a real upper quasi-triangular matrix  $T$  in canonical Schur form. These are the same as the condition numbers of the eigenvalues and right eigenvectors of an original matrix  $A = ZTZ^T$  (with orthogonal  $Z$ ), from which  $T$  may have been derived.

F08QLF computes the reciprocal of the condition number of an eigenvalue  $\lambda_i$  as

$$s_i = \frac{|v^H u|}{\|u\|_E \|v\|_E},$$

where  $u$  and  $v$  are the right and left eigenvectors of  $T$ , respectively, corresponding to  $\lambda_i$ . This reciprocal condition number always lies between zero (i.e., ill-conditioned) and one (i.e., well-conditioned).

An approximate error estimate for a computed eigenvalue  $\lambda_i$  is then given by

$$\frac{\epsilon \|T\|}{s_i},$$

where  $\epsilon$  is the *machine precision*.

To estimate the reciprocal of the condition number of the right eigenvector corresponding to  $\lambda_i$ , the routine first calls F08QFF (STREXC/DTREXC) to reorder the eigenvalues so that  $\lambda_i$  is in the leading position:

$$T = Q \begin{pmatrix} \lambda_i & c^T \\ 0 & T_{22} \end{pmatrix} Q^T.$$

The reciprocal condition number of the eigenvector is then estimated as  $sep_i$ , the smallest singular value of the matrix  $(T_{22} - \lambda_i I)$ . This number ranges from zero (i.e., ill-conditioned) to very large (i.e., well-conditioned).

An approximate error estimate for a computed right eigenvector  $u$  corresponding to  $\lambda_i$  is then given by

$$\frac{\epsilon \|T\|}{sep_i}.$$

## 4 References

- [1] Golub G H and van Loan C F (1996) *Matrix Computations* Johns Hopkins University Press (3rd Edition), Baltimore

## 5 Parameters

- 1:** JOB — CHARACTER\*1 *Input*  
*On entry:* indicates whether condition numbers are required for eigenvalues and/or eigenvectors, as follows:  
 if JOB = 'E', then condition numbers for eigenvalues only are computed;  
 if JOB = 'V', then condition numbers for eigenvectors only are computed;  
 if JOB = 'B', then condition numbers for both eigenvalues and eigenvectors are computed.  
*Constraint:* JOB = 'E', 'V' or 'B'.
- 2:** HOWMNY — CHARACTER\*1 *Input*  
*On entry:* indicates how many condition numbers are to be computed, as follows:  
 if HOWMNY = 'A', then condition numbers for all eigenpairs are computed;  
 if HOWMNY = 'S', then condition numbers for selected eigenpairs (as specified by SELECT) are computed.  
*Constraint:* HOWMNY = 'A' or 'S'.
- 3:** SELECT(\*) — LOGICAL array *Input*  
**Note:** the dimension of the array SELECT must be at least  $\max(1, N)$  if HOWMNY = 'S' and at least 1 otherwise.  
*On entry:* SELECT specifies the eigenpairs for which condition numbers are to be computed if HOWMNY = 'S'. To select condition numbers for the eigenpair corresponding to the real eigenvalue  $\lambda_j$ , SELECT( $j$ ) must be set .TRUE.. To select condition numbers corresponding to a complex conjugate pair of eigenvalues  $\lambda_j$  and  $\lambda_{j+1}$ , SELECT( $j$ ) and/or SELECT( $j+1$ ) must be set .TRUE..  
 SELECT is not referenced if HOWMNY = 'A'.
- 4:** N — INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $T$ .  
*Constraint:*  $N \geq 0$ .
- 5:** T(LDT,\*) — *real* array *Input*  
**Note:** the second dimension of the array T must be at least  $\max(1, N)$ .  
*On entry:* the  $n$  by  $n$  upper quasi-triangular matrix  $T$  in canonical Schur form, as returned by F08PEF (SHSEQR/DHSEQR).
- 6:** LDT — INTEGER *Input*  
*On entry:* the first dimension of the array T as declared in the (sub)program from which F08QLF (STRSNA/DTRSNA) is called.  
*Constraint:*  $LDT \geq \max(1, N)$ .
- 7:** VL(LDVL,\*) — *real* array *Input*  
**Note:** the second dimension of the array VL must be at least  $\max(1, MM)$  if JOB = 'E' or 'B' and at least 1 if JOB = 'V'.  
*On entry:* if JOB = 'E' or 'B', VL must contain the left eigenvectors of  $T$  (or of any matrix  $QTQ^T$  with  $Q$  orthogonal) corresponding to the eigenpairs specified by HOWMNY and SELECT. The eigenvectors **must** be stored in consecutive columns of VL, as returned by F08QKF (STREVC/DTREVC) or F08PKF (SHSEIN/DHSEIN).  
 VL is not referenced if JOB = 'V'.

- 8:** LDVL — INTEGER *Input*  
*On entry:* the first dimension of the array VL as declared in the (sub)program from which F08QLF (STRSNA/DTRSNA) is called.  
*Constraints:*  
 $LDVL \geq \max(1, N)$  if JOB = 'E' or 'B',  
 $LDVL \geq 1$  if JOB = 'V'.
- 9:** VR(LDVR,\*) — *real* array *Input*  
**Note:** the second dimension of the array VR must be at least  $\max(1, MM)$  if JOB = 'E' or 'B' and at least 1 if JOB = 'V'.  
*On entry:* if JOB = 'E' or 'B', VR must contain the right eigenvectors of  $T$  (or of any matrix  $QTQ^T$  with  $Q$  orthogonal) corresponding to the eigenpairs specified by HOWMNY and SELECT. The eigenvectors **must** be stored in consecutive columns of VR, as returned by F08QKF (STREVC/DTREVC) or F08PKF (SHSEIN/DHSEIN).  
 VR is not referenced if JOB = 'V'.
- 10:** LDVR — INTEGER *Input*  
*On entry:* the first dimension of the array VR as declared in the (sub)program from which F08QLF (STRSNA/DTRSNA) is called.  
*Constraints:*  
 $LDVR \geq \max(1, N)$  if JOB = 'E' or 'B',  
 $LDVR \geq 1$  if JOB = 'V'.
- 11:** S(\*) — *real* array *Output*  
**Note:** the dimension of the array S must be at least  $\max(1, MM)$  if JOB = 'E' or 'B' and at least 1 if JOB = 'V'.  
*On exit:* the reciprocal condition numbers of the selected eigenvalues if JOB = 'E' or 'B', stored in consecutive elements of the array. Thus  $S(j)$ ,  $SEP(j)$  and the  $j$ th columns of VL and VR all correspond to the same eigenpair (but not in general the  $j$ th eigenpair unless all eigenpairs have been selected). For a complex conjugate pair of eigenvalues, two consecutive elements of S are set to the same value.  
 S is not referenced if JOB = 'V'.
- 12:** SEP(\*) — *real* array *Output*  
**Note:** the dimension of the array SEP must be at least  $\max(1, MM)$  if JOB = 'V' or 'B' and at least 1 if JOB = 'E'.  
*On exit:* the estimated reciprocal condition numbers of the selected right eigenvectors if JOB = 'V' or 'B', stored in consecutive elements of the array. For a complex eigenvector, two consecutive elements of SEP are set to the same value. If the eigenvalues cannot be reordered to compute  $SEP(j)$ , then  $SEP(j)$  is set to zero; this can only occur when the true value would be very small anyway.  
 SEP is not referenced if JOB = 'E'.
- 13:** MM — INTEGER *Input*  
*On entry:* the number of elements in the arrays S and SEP, and the number of columns in the arrays VL and VR (if used). The precise number required,  $m$ , is  $n$  if HOWMNY = 'A'; if HOWMNY = 'S',  $m$  is obtained by counting 1 for each selected real eigenvalue, and 2 for each selected complex conjugate pair of eigenvalues (see SELECT), in which case  $0 \leq m \leq n$ .  
*Constraint:*  $MM \geq M$ .

- 14:** M — INTEGER *Output*  
*On exit:*  $m$ , the number of elements of S and/or SEP actually used to store the estimated condition numbers. If HOWMNY = 'A', M is set to  $n$ .
- 15:** WORK(LDWORK,\*) — *real* array *Workspace*  
**Note:** the second dimension of the array WORK must be at least  $\max(1, N+6)$  if JOB = 'V' or 'B' and at least 1 if JOB = 'E'.  
 WORK is not referenced if JOB = 'E'.
- 16:** LDWORK — INTEGER *Input*  
*On entry:* the first dimension of the array WORK as declared in the (sub)program from which F08QLF (STRSNA/DTRSNA) is called.  
*Constraints:*  
 $\text{LDWORK} \geq \max(1, N)$  if JOB = 'V' or 'B',  
 $\text{LDWORK} \geq 1$  if JOB = 'E'.
- 17:** IWORK(\*) — INTEGER array *Workspace*  
**Note:** the dimension of the array IWORK must be at least  $\max(1, 2*(N-1))$ .
- 18:** INFO — INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed values  $sep_i$  may overestimate the true value, but seldom by a factor of more than 3.

## 8 Further Comments

For a description of canonical Schur form, see the document for F08PEF (SHSEQR/DHSEQR).

The complex analogue of this routine is F08QYF (CTRSNA/ZTRSNA).

## 9 Example

To compute approximate error estimates for all the eigenvalues and right eigenvectors of the matrix  $T$ , where

$$T = \begin{pmatrix} 0.7995 & -0.1144 & 0.0060 & 0.0336 \\ 0.0000 & -0.0994 & 0.2478 & 0.3474 \\ 0.0000 & -0.6483 & -0.0994 & 0.2026 \\ 0.0000 & 0.0000 & 0.0000 & -0.1007 \end{pmatrix}.$$

## 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*   F08QLF Example Program Text
*   Mark 16 Release. NAG Copyright 1992.
*   .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX, LDT, LDWORK, LDVL, LDVR
PARAMETER       (NMAX=8,LDT=NMAX,LDWORK=NMAX,LDVL=NMAX,LDVR=NMAX)
*   .. Local Scalars ..
real           EPS, TNORM
INTEGER          I, INFO, J, M, N
*   .. Local Arrays ..
real           S(NMAX), SEP(NMAX), T(LDT,NMAX), VL(LDVL,NMAX),
+              VR(LDVR,NMAX), WORK(LDWORK,NMAX+6)
INTEGER          IWORK(2*NMAX-2)
LOGICAL         SELECT(1)
*   .. External Functions ..
real           F06RAF, X02AJF
EXTERNAL        F06RAF, X02AJF
*   .. External Subroutines ..
EXTERNAL        strevc, strsna
*   .. Executable Statements ..
WRITE (NOUT,*) 'F08QLF Example Program Results'
Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*   Read T from data file
*
READ (NIN,*) ((T(I,J),J=1,N),I=1,N)
*
*   Calculate the left and right eigenvectors of T
*
CALL strevc('Both','All',SELECT,N,T,LDT,VL,LDVL,VR,LDVR,N,M,
+          WORK,INFO)
*
*   Estimate condition numbers for all the eigenvalues and right
*   eigenvectors of T
*
CALL strsna('Both','All',SELECT,N,T,LDT,VL,LDVL,VR,LDVR,S,SEP,
+          N,M,WORK,LDWORK,IWORK,INFO)
*
*   Print condition numbers of eigenvalues and right eigenvectors
*
WRITE (NOUT,*)
WRITE (NOUT,*) 'S'
WRITE (NOUT,99999) (S(I),I=1,M)
WRITE (NOUT,*)
WRITE (NOUT,*) 'SEP'
WRITE (NOUT,99999) (SEP(I),I=1,M)
*
*   Calculate approximate error estimates (using the 1-norm)
*
EPS = X02AJF()

```

```

      TNORM = F06RAF('1-norm',N,N,T,LDT,WORK)
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Approximate error estimates for eigenvalues ',
+   'of T (machine-dependent)'
      WRITE (NOUT,99999) (EPS*TNORM/S(I),I=1,M)
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Approximate error estimates for right ',
+   'eigenvectors of T (machine-dependent)'
      WRITE (NOUT,99999) (EPS*TNORM/SEP(I),I=1,M)
      END IF
      STOP
*
99999 FORMAT ((3X,1P,7e11.1))
      END

```

## 9.2 Program Data

F08QLF Example Program Data

```

4                                     :Value of N
0.7995  -0.1144  0.0060  0.0336
0.0000  -0.0994  0.2478  0.3474
0.0000  -0.6483  -0.0994  0.2026
0.0000  0.0000  0.0000  -0.1007 :End of matrix T

```

## 9.3 Program Results

F08QLF Example Program Results

```

S
   9.9E-01   7.0E-01   7.0E-01   5.7E-01

SEP
   6.3E-01   3.7E-01   3.7E-01   3.1E-01

Approximate error estimates for eigenvalues of T (machine-dependent)
   9.6E-17   1.4E-16   1.4E-16   1.7E-16

Approximate error estimates for right eigenvectors of T (machine-dependent)
   1.5E-16   2.6E-16   2.6E-16   3.1E-16

```

---